

```

10 '*****
20 '          RESOLUTION D'EQUATIONS          *
30 '          M. ROUSSELET                    *
40 '*****
50 MODE 2:PRINT CHR$(30);CHR$(23);CHR$(0
)
60 'definition des fonctions f et g
70 DEF FN f(x)=x*x*x+3*x+1
80 DEF FN g(x)=5*x*COS(x)
90 'definition de xmin et de xmax
100 INPUT"quelle est la valeur de xmin";
xmin
110 INPUT"quelle est la valeur de xmax";
xmax

```

```

120 IF xmin>=xmax THEN PRINT "intervalle
    mal choisi":GOTO 100
130 n=50
140 CLS
150 'definition et dessin de la fenetre
160 umin=50:vmin=50:umax=500:vmax=350
170 PLOT umin,vmin,1
180 DRAW umax,vmin,1
190 DRAW umax,vmax,1
200 DRAW umin,vmax,1
210 DRAW umin,vmin,1
220 'division de l'intervalle
    [xmin,xmax] en parties egales
    puis calcul de x(i),f(x(i),g(xi)),
    ymin et ymax
230 DIM x(1+n):DIM y(1+n):DIM z(1+n)
240 ymax=-1E+30:ymin=1E+30
250 FOR i=0 TO n
260 x(i)=xmin+i*(xmax-xmin)/n
270 y(i)=FN f(x(i))
280 z(i)=FN g(x(i))
290 IF y(i)>ymax THEN ymax=y(i)
300 IF z(i)>ymax THEN ymax=z(i)
310 IF y(i)<ymin THEN ymin=y(i)

320 IF z(i)<ymin THEN ymin=z(i)
330 NEXT i
340 'adaptation a la taille d'ecran
350 k1=(umax-umin)/(xmax-xmin):l1=umax-k
    1*xmax
360 k2=(vmax-vmin)/(ymax-ymin):l2=vmax-k
    2*ymax
370 FOR i=0 TO n
380 x(i)=k1*x(i)+l1:y(i)=k2*y(i)+l2:z(i)
    =k2*z(i)+l2
390 NEXT i
400 'dessin des 2 courbes
410 PRINT CHR$(30);CHR$(23);CHR$(1)
420 FOR i=0 TO n-1
430 PLOT x(i),y(i):DRAW x(i+1),y(i+1)
440 PLOT x(i),z(i):DRAW x(i+1),z(i+1)
450 NEXT i
460 LOCATE 10,24
470 PRINT "xmin=";xmin,"xmax=";xmax
480 'recherche de l'abscisse du point
    d'intersection par deplacement
    du curseur
490 x=300:y=200:MOVE x-10,y:DRAWR 20,0:M
    OVE x,y-10:DRAWR 0,20
500 a$="":WHILE a$="":a$=INKEY$:WEND
510 IF TESTR(0,0) THEN MOVE x-10,y:DRAWR
    20,0:MOVE x,y-10:DRAWR 0,20
520 IF a$=CHR$(243) THEN x=x+1
530 IF a$=CHR$(242) THEN x=x-1
540 IF a$=CHR$(240) THEN y=y+1
550 IF a$=CHR$(241) THEN y=y-1
560 IF a$="s" OR a$="S" THEN 610
570 IF a$="r" OR a$="R" THEN RUN
580 MOVE x-10,y:DRAWR 20,0:MOVE x,y-10:D
    RAWR 0,20
590 GOTO 500
600 'affichage de la solution
610 LOCATE 10,24:sol=(x-l1)/k1
620 PRINT"solution: X=";sol
630 PRINT "verification:"
640 PRINT "f(x)=";FN f(sol)
650 PRINT "g(x)=";FN g(sol)
660 END

```